Problem 7.5

Enter all the constants in consistent units (or let Mathematica do all the unit conversions).

In[1]:=
```
k = Quantity[1, "BoltzmannConstant"];
h = Quantity[1, "PlanckConstant"];
m = Quantity[1, "ElectronMass"];
Ie = Quantity[0.044, "Electronvolts"] (* Ionization energy *);
nd = Quantity[10^17, 1/"Centimeters"^3] (* Donor density *);
```

Enter the dimensionless helper function. Assume we will enter temperatures in Kelvin.

In[6]:=
```
y[T_] := nd (h^3/(2 π m k Quantity[T, "Kelvins"])^(3/2)) Exp[Ie/(k Quantity[T, "Kelvins"])]
```

In[7]:=
```
x[T_] := 1/(2 y[T]) (Sqrt[1 + 4 y[T]] - 1)
```

In[8]:=
```
x[T]
```

Out[8]=

$$0.0120734152\, e^{-510.599/T} \left(-1 + \sqrt{1 + \frac{165.653211\, e^{510.599/T}}{T^{3/2}}}\right) T^{3/2}$$

In[9]:=
```
Plot[x[T], {T, 5, 300}, LabelStyle → Larger, AxesLabel → {"T (K)", "Nc/Nd"},
 ImageSize → Scaled[0.8], PlotLabel → "Fraction of Ionized Donors"]
```

Out[9]=



At room temperature, nearly all of the donors will be ionized. The temperature at which half of them will be ionized is about 122 K:

In[∘]:=
```
FindRoot[x[T] == 0.5, {T, 50}]
```
Out[∘]=
```
{T → 122.208}
```

Further explorations: Chemical Potential

The chemical potential for Nd donated electrons is

In[11]:=
```
lQ[T_] := h / Sqrt[2 π m k T]
```

In[34]:=
```
UnitConvert[lQ[Quantity[300, "Kelvins"]], "Nanometers"]
```
Out[34]=
```
4.30347544 nm
```

In[12]:=
```
vQ[T_] := lQ[T]^3
```

In[31]:=
```
(* Assuming Nc = Nd (i.e. every donor is ionized *)
μ[T_] := UnitConvert[
  -k Quantity[T, "Kelvins"] Log[2 / (nd vQ[Quantity[T, "Kelvins"]])], "Electronvolts"]
```

In[32]:=
```
Plot[{-Ie, μ[T]}, {T, 50, 200}]
```
Out[32]=